Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
wdir = SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
NotebookOpen[wdir <> "\\MakeVSnips.nb"];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization

The "degree carrier / filtration parameter" is $\hbar$, and all "coupling constants" are proportional to it.

TD

```
$TħD = 4; ħ /: ħ^d_. /; d > $TħD := 0;
$TϵD = 2; ϵ /: ϵ^d_. /; d > $TϵD := 0;
SetAttributes[SS, HoldAll];
SS[ℰ_] := Block[{ħ, ϵ}, (* Shielded Series *)
  Collect[Normal@Series[ℰ /. {T_i_ → ⅇ^(ħ t_i/2), T → ⅇ^(ħ t/2)}, {ħ, 0, $TħD}], ħ, Together]]
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply]; Attributes[NonCommutativeMultiply] = {};
(NCM = NonCommutativeMultiply)[x_] := x;
NCM[x_, y_, z__] := (x ** y) ** z;
0 ** _ = _ ** 0 = 0;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, CE, pow,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  CE[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp :→ t_i, u_U :→ Replace[u, x_ :→ x_i, 1]};
  U_i_[NCM[]] := U[];
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := CE[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, CE[a b (x ** y)]];
  (a_ * x_U) ** y_ := CE[a (x ** y)]; x_ ** (a_ * y_U) := CE[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := CE[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := CE[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := CE[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := CE[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us},
    sp = Replace[{specs}, l_List :→ (l → null), {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. (l_ → s_) :→ (l /. x_i_ :→ x_s));
    CE[Total[
        CoefficientRules[poly, vs] /. (p_ → c_) :→ c U@(us^p)
      ]] /. x_null :→ x
    ];
  pow[ℰ_, 0] = U[]; pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
  𝕊_U[ℰ_, ss___Rule] := CE@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) :→ c NCM @@ MapThread[pow, {Last /@ {ss}, p}]];
  S_i_[c_. * u_U] := CE[(c /. S_i[U, Centrals]) DeleteCases[u, _i] **
      U_i[NCM @@ Reverse@Cases[u, x_i :→ S@U@x]]];
 ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) :→ (m[U[g]] = img), {1}];
  m[U[]] = V[];
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NCM @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U :→ m[u]]; )
```

Meta-Operations

QLImplementation

```
S_i_[ℰ_Plus] := Simp[S_i /@ ℰ];
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
(S@CU@y = -CU@y; S@CU@a = -CU@a; S@CU@x = -CU@x;)
S_i_[CU, Centrals] = {t_i → -t_i};
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

$\{0.625, \{(28\, t^2 \gamma^4 + 116\, t\, \gamma^5 \epsilon + 120\, \gamma^6 \epsilon^2)\, \mathrm{CU}[y, y, y, x, x] +$
$\ll 23 \gg + \mathrm{CU}[y, y, y, y, y, a, a, a, a, x, x, x, x, x], 0\}\}$

```
S_1[CU[y_1, a_1, x_1]]
```

$-2 \epsilon\, \mathrm{CU}[a_1, a_1] + 2 \gamma\, \mathrm{CU}[y_1, x_1] - \mathrm{CU}[y_1, a_1, x_1] - \gamma\, \mathrm{CU}[\,]\, t_1 + \mathrm{CU}[a_1]\, (2 \gamma \epsilon + t_1)$

Verifying that *S* is an anti-homomorphism on CU:

```
With[{bas = CU /@ {y₁, a₁, x₁}},
 Table[HL@Simp[S₁[z1 ** z2] - S₁[z2] ** S₁[z1]],
  {z1, bas}, {z2, bas} ] ]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying the involutivity of S on products of triples:

```
With[{bas = CU /@ {y₁, a₁, x₁}},
 Table[HL@Simp[z1 ** z2 ** z3 - S₁@S₁[z1 ** z2 ** z3]],
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, $T = e^{\hbar t/2}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle/\mathcal{R}$, where $\mathcal{R} = \left([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = \left(1 - T^2 A^2\right)/\hbar,\ [x, a] = -\gamma x\right)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q = SS[e^(γ ϵ ħ)]; (*T=SS[e^(ħ t/2)];*)
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] = (q - 1) QU@{y, x} + 𝕆_QU[SS[(1 - T^2 e^(-2 ϵ a ħ)) / ħ], {a}];
(S@QU@y = 𝕆_QU[SS[-T^-2 e^(ħ ϵ a) y], {a, y}];
 S@QU@a = -QU@a; S@QU@x = 𝕆_QU[SS[-e^(ħ ϵ a) x], {a, x}];)
S_i_[QU, Centrals] = {t_i → -t_i, T_i → T_i^-1};
```

```
With[{bas = QU /@ {y, a, x}}, Table[{z1, z2} → Simp[z1 ** z2 - z2 ** z1], {z1, bas}, {z2, bas}] ]
```

$$\Big\{\{\{QU[y], QU[y]\} \to 0,\ \{QU[y], QU[a]\} \to \gamma\, QU[y],$$
$$\{QU[y], QU[x]\} \to \left(t + \frac{t^2\hbar}{2} + \frac{t^3\hbar^2}{6} + \frac{t^4\hbar^3}{24} + \frac{t^5\hbar^4}{120} + \frac{t^6\hbar^5}{720}\right) QU[\,] +$$
$$\left(-2\epsilon - 2t\epsilon\hbar - t^2\epsilon\hbar^2 - \frac{1}{3}t^3\epsilon\hbar^3 - \frac{1}{12}t^4\epsilon\hbar^4 - \frac{1}{60}t^5\epsilon\hbar^5\right) QU[a] +$$
$$\left(2\epsilon^2\hbar + 2t\epsilon^2\hbar^2 + t^2\epsilon^2\hbar^3 + \frac{1}{3}t^3\epsilon^2\hbar^4 + \frac{1}{12}t^4\epsilon^2\hbar^5\right) QU[a, a] + \left(-\gamma\epsilon\hbar - \frac{1}{2}\gamma^2\epsilon^2\hbar^2\right) QU[y, x]\},$$
$$\{\{QU[a], QU[y]\} \to -\gamma\, QU[y],\ \{QU[a], QU[a]\} \to 0,\ \{QU[a], QU[x]\} \to \gamma\, QU[x]\},$$
$$\{\{QU[x], QU[y]\} \to \left(-t - \frac{t^2\hbar}{2} - \frac{t^3\hbar^2}{6} - \frac{t^4\hbar^3}{24} - \frac{t^5\hbar^4}{120} - \frac{t^6\hbar^5}{720}\right) QU[\,] +$$
$$\left(2\epsilon + 2t\epsilon\hbar + t^2\epsilon\hbar^2 + \frac{1}{3}t^3\epsilon\hbar^3 + \frac{1}{12}t^4\epsilon\hbar^4 + \frac{1}{60}t^5\epsilon\hbar^5\right) QU[a] +$$
$$\left(-2\epsilon^2\hbar - 2t\epsilon^2\hbar^2 - t^2\epsilon^2\hbar^3 - \frac{1}{3}t^3\epsilon^2\hbar^4 - \frac{1}{12}t^4\epsilon^2\hbar^5\right) QU[a, a] + \left(\gamma\epsilon\hbar + \frac{1}{2}\gamma^2\epsilon^2\hbar^2\right) QU[y, x],$$
$$\{QU[x], QU[a]\} \to -\gamma\, QU[x],\ \{QU[x], QU[x]\} \to 0\}\Big\}$$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a "random" triple (~34 secs @ $TℏD=5, $TϵD=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp ]
  }] // Timing
```

$$\left\{50.7031, \left\{\left(28\, t^2 \gamma^4 + 116\, t\, \gamma^5 \epsilon + 120\, \gamma^6 \epsilon^2 + \ll 12 \gg + \frac{7}{10} t^7 \gamma^4 \hbar^5 + \frac{1549}{90} t^6 \gamma^5 \epsilon \hbar^5 + \frac{2215}{12} t^5 \gamma^6 \epsilon^2 \hbar^5\right) \mathrm{QU}[y, \ll 3 \gg, x] + \ll 22 \gg, 0\right\}\right\}$$

Verifying that *S* is an anti-homomorphism on QU:

```
S₁[QU@x₁]
```

$$-\mathrm{QU}[x_1] - \epsilon \hbar\, \mathrm{QU}[a_1, x_1] - \frac{1}{2} \epsilon^2 \hbar^2\, \mathrm{QU}[a_1, a_1, x_1]$$

```
S₁[QU@y₁]
```

$$\mathrm{QU}[y_1, a_1, a_1] \left(-\frac{1}{2} \epsilon^2 \hbar^2 + \frac{1}{2} \epsilon^2 \hbar^3 t_1 - \frac{1}{4} \epsilon^2 \hbar^4 t_1^2 + \frac{1}{12} \epsilon^2 \hbar^5 t_1^3\right) + \mathrm{QU}[y_1, a_1] \left(-\epsilon \hbar + \gamma \epsilon^2 \hbar^2 + \epsilon \hbar^2 t_1 - \gamma \epsilon^2 \hbar^3 t_1 - \frac{1}{2} \epsilon \hbar^3 t_1^2 + \frac{1}{2} \gamma \epsilon^2 \hbar^4 t_1^2 + \frac{1}{6} \epsilon \hbar^4 t_1^3 - \frac{1}{6} \gamma \epsilon^2 \hbar^5 t_1^3 - \frac{1}{24} \epsilon \hbar^5 t_1^4\right) + \mathrm{QU}[y_1] \left(-1 + \gamma \epsilon \hbar - \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2 + \hbar t_1 - \gamma \epsilon \hbar^2 t_1 + \frac{1}{2} \gamma^2 \epsilon^2 \hbar^3 t_1 - \frac{1}{2} \hbar^2 t_1^2 + \frac{1}{2} \gamma \epsilon \hbar^3 t_1^2 - \frac{1}{4} \gamma^2 \epsilon^2 \hbar^4 t_1^2 + \frac{1}{6} \hbar^3 t_1^3 - \frac{1}{6} \gamma \epsilon \hbar^4 t_1^3 + \frac{1}{12} \gamma^2 \epsilon^2 \hbar^5 t_1^3 - \frac{1}{24} \hbar^4 t_1^4 + \frac{1}{24} \gamma \epsilon \hbar^5 t_1^4 + \frac{1}{120} \hbar^5 t_1^5\right)$$

```
{z1, z2} = {QU[x₁], QU[y₁]}; z1 ** z2
```

$$\mathrm{QU}[y_1, x_1] + \left(\gamma \epsilon \hbar + \frac{1}{2} \gamma^2 \epsilon^2 \hbar^2\right) \mathrm{QU}[y_1, x_1] + \mathrm{QU}[a_1, a_1] \left(-2 \epsilon^2 \hbar - 2 \epsilon^2 \hbar^2 t_1 - \epsilon^2 \hbar^3 t_1^2 - \frac{1}{3} \epsilon^2 \hbar^4 t_1^3 - \frac{1}{12} \epsilon^2 \hbar^5 t_1^4\right) + \mathrm{QU}[a_1] \left(2 \epsilon + 2 \epsilon \hbar t_1 + \epsilon \hbar^2 t_1^2 + \frac{1}{3} \epsilon \hbar^3 t_1^3 + \frac{1}{12} \epsilon \hbar^4 t_1^4 + \frac{1}{60} \epsilon \hbar^5 t_1^5\right) + \mathrm{QU}[] \left(-t_1 - \frac{\hbar t_1^2}{2} - \frac{1}{6} \hbar^2 t_1^3 - \frac{1}{24} \hbar^3 t_1^4 - \frac{1}{120} \hbar^4 t_1^5 - \frac{1}{720} \hbar^5 t_1^6\right)$$

S_1[z1 ** z2]

$$\mathrm{QU}[y_1, a_1, a_1, x_1]\left(2\,\epsilon^2\hbar^2 - 2\,\epsilon^2\hbar^3 t_1 + \epsilon^2\hbar^4 t_1^2 - \frac{1}{3}\epsilon^2\hbar^5 t_1^3\right) + \mathrm{QU}[y_1, a_1, x_1]\left(2\,\epsilon\,\hbar - 2\,\gamma\,\epsilon^2\hbar^2 - 2\,\epsilon\,\hbar^2 t_1 + 2\,\gamma\,\epsilon^2\hbar^3 t_1 + \epsilon\,\hbar^3 t_1^2 - \gamma\,\epsilon^2\hbar^4 t_1^2 - \frac{1}{3}\epsilon\,\hbar^4 t_1^3 + \frac{1}{3}\gamma\,\epsilon^2\hbar^5 t_1^3 + \frac{1}{12}\epsilon\,\hbar^5 t_1^4\right) + \mathrm{QU}[y_1, x_1]\left(1 - \gamma\,\epsilon\,\hbar + \frac{1}{2}\gamma^2\epsilon^2\hbar^2 - \hbar\,t_1 + \gamma\,\epsilon\,\hbar^2 t_1 - \frac{1}{2}\gamma^2\epsilon^2\hbar^3 t_1 + \frac{1}{2}\hbar^2 t_1^2 - \frac{1}{2}\gamma\,\epsilon\,\hbar^3 t_1^2 + \frac{1}{4}\gamma^2\epsilon^2\hbar^4 t_1^2 - \frac{1}{6}\hbar^3 t_1^3 + \frac{1}{6}\gamma\,\epsilon\,\hbar^4 t_1^3 - \frac{1}{12}\gamma^2\epsilon^2\hbar^5 t_1^3 + \frac{1}{24}\hbar^4 t_1^4 - \frac{1}{24}\gamma\,\epsilon\,\hbar^5 t_1^4 - \frac{1}{120}\hbar^5 t_1^5\right)$$

S_1@QU@{x_1, y_1}

$$\mathrm{QU}[y_1, a_1, a_1, x_1]\left(2\,\epsilon^2\hbar^2 - 2\,\epsilon^2\hbar^3 t_1 + \epsilon^2\hbar^4 t_1^2 - \frac{1}{3}\epsilon^2\hbar^5 t_1^3\right) + \mathrm{QU}[y_1, a_1, x_1]\left(2\,\epsilon\,\hbar - 2\,\gamma\,\epsilon^2\hbar^2 - 2\,\epsilon\,\hbar^2 t_1 + 2\,\gamma\,\epsilon^2\hbar^3 t_1 + \epsilon\,\hbar^3 t_1^2 - \gamma\,\epsilon^2\hbar^4 t_1^2 - \frac{1}{3}\epsilon\,\hbar^4 t_1^3 + \frac{1}{3}\gamma\,\epsilon^2\hbar^5 t_1^3 + \frac{1}{12}\epsilon\,\hbar^5 t_1^4\right) + \mathrm{QU}[y_1, x_1]\left(1 - \gamma\,\epsilon\,\hbar + \frac{1}{2}\gamma^2\epsilon^2\hbar^2 - \hbar\,t_1 + \gamma\,\epsilon\,\hbar^2 t_1 - \frac{1}{2}\gamma^2\epsilon^2\hbar^3 t_1 + \frac{1}{2}\hbar^2 t_1^2 - \frac{1}{2}\gamma\,\epsilon\,\hbar^3 t_1^2 + \frac{1}{4}\gamma^2\epsilon^2\hbar^4 t_1^2 - \frac{1}{6}\hbar^3 t_1^3 + \frac{1}{6}\gamma\,\epsilon\,\hbar^4 t_1^3 - \frac{1}{12}\gamma^2\epsilon^2\hbar^5 t_1^3 + \frac{1}{24}\hbar^4 t_1^4 - \frac{1}{24}\gamma\,\epsilon\,\hbar^5 t_1^4 - \frac{1}{120}\hbar^5 t_1^5\right)$$

S_1@QU[y_1, x_1]

$$\mathrm{QU}[y_1, a_1, a_1, x_1]\left(2\,\epsilon^2\hbar^2 - 2\,\epsilon^2\hbar^3 t_1 + \epsilon^2\hbar^4 t_1^2 - \frac{1}{3}\epsilon^2\hbar^5 t_1^3\right) + \mathrm{QU}[y_1, a_1, x_1]\left(2\,\epsilon\,\hbar - 4\,\gamma\,\epsilon^2\hbar^2 - 2\,\epsilon\,\hbar^2 t_1 + 4\,\gamma\,\epsilon^2\hbar^3 t_1 + \epsilon\,\hbar^3 t_1^2 - 2\,\gamma\,\epsilon^2\hbar^4 t_1^2 - \frac{1}{3}\epsilon\,\hbar^4 t_1^3 + \frac{2}{3}\gamma\,\epsilon^2\hbar^5 t_1^3 + \frac{1}{12}\epsilon\,\hbar^5 t_1^4\right) + \mathrm{QU}[a_1, a_1]\left(2\,\epsilon^2\hbar - 2\,\epsilon^2\hbar^2 t_1 + \epsilon^2\hbar^3 t_1^2 - \frac{1}{3}\epsilon^2\hbar^4 t_1^3 + \frac{1}{12}\epsilon^2\hbar^5 t_1^4\right) + \mathrm{QU}[y_1, x_1]\left(1 - 2\,\gamma\,\epsilon\,\hbar + 2\,\gamma^2\epsilon^2\hbar^2 - \hbar\,t_1 + 2\,\gamma\,\epsilon\,\hbar^2 t_1 - 2\,\gamma^2\epsilon^2\hbar^3 t_1 + \frac{1}{2}\hbar^2 t_1^2 - \gamma\,\epsilon\,\hbar^3 t_1^2 + \gamma^2\epsilon^2\hbar^4 t_1^2 - \frac{1}{6}\hbar^3 t_1^3 + \frac{1}{3}\gamma\,\epsilon\,\hbar^4 t_1^3 - \frac{1}{3}\gamma^2\epsilon^2\hbar^5 t_1^3 + \frac{1}{24}\hbar^4 t_1^4 - \frac{1}{12}\gamma\,\epsilon\,\hbar^5 t_1^4 - \frac{1}{120}\hbar^5 t_1^5\right) + \mathrm{QU}[a_1]\left(2\,\epsilon - 2\,\gamma\,\epsilon^2\hbar - 2\,\epsilon\,\hbar\,t_1 + 2\,\gamma\,\epsilon^2\hbar^2 t_1 + \epsilon\,\hbar^2 t_1^2 - \gamma\,\epsilon^2\hbar^3 t_1^2 - \frac{1}{3}\epsilon\,\hbar^3 t_1^3 + \frac{1}{3}\gamma\,\epsilon^2\hbar^4 t_1^3 + \frac{1}{12}\epsilon\,\hbar^4 t_1^4 - \frac{1}{12}\gamma\,\epsilon^2\hbar^5 t_1^4 - \frac{1}{60}\epsilon\,\hbar^5 t_1^5\right) + \mathrm{QU}[\,]\left(-t_1 + \gamma\,\epsilon\,\hbar\,t_1 - \frac{1}{2}\gamma^2\epsilon^2\hbar^2 t_1 + \frac{\hbar\,t_1^2}{2} - \frac{1}{2}\gamma\,\epsilon\,\hbar^2 t_1^2 + \frac{1}{4}\gamma^2\epsilon^2\hbar^3 t_1^2 - \frac{1}{6}\hbar^2 t_1^3 + \frac{1}{6}\gamma\,\epsilon\,\hbar^3 t_1^3 - \frac{1}{12}\gamma^2\epsilon^2\hbar^4 t_1^3 + \frac{1}{24}\hbar^3 t_1^4 - \frac{1}{24}\gamma\,\epsilon\,\hbar^4 t_1^4 + \frac{1}{48}\gamma^2\epsilon^2\hbar^5 t_1^4 - \frac{1}{120}\hbar^4 t_1^5 + \frac{1}{120}\gamma\,\epsilon\,\hbar^5 t_1^5 + \frac{1}{720}\hbar^5 t_1^6\right)$$

```
With[{bas = QU /@ {y₁, a₁, x₁}},
 Table[{z1, z2} → HL@Simp[S₁[z1 ** z2] - S₁[z2] ** S₁[z1]],
  {z1, bas}, {z2, bas}]]
```

$$\{\{\{\mathrm{QU}[y_1], \mathrm{QU}[y_1]\} \to 0, \{\mathrm{QU}[y_1], \mathrm{QU}[a_1]\} \to 0, \{\mathrm{QU}[y_1], \mathrm{QU}[x_1]\} \to 0\},$$
$$\{\{\mathrm{QU}[a_1], \mathrm{QU}[y_1]\} \to 0, \{\mathrm{QU}[a_1], \mathrm{QU}[a_1]\} \to 0, \{\mathrm{QU}[a_1], \mathrm{QU}[x_1]\} \to 0\},$$
$$\{\{\mathrm{QU}[x_1], \mathrm{QU}[y_1]\} \to 0, \{\mathrm{QU}[x_1], \mathrm{QU}[a_1]\} \to 0, \{\mathrm{QU}[x_1], \mathrm{QU}[x_1]\} \to 0\}\}$$

Verifying that $\text{Lim}_{\hbar\to 0}$ QU = CU using a “random” product (~23 secs @ \$T$\hbar$D=5, \$TϵD=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → ℯ^(ℏ t/2)}, ℏ → 0] - lhs] // HL
  }] // Timing
```

```
{32.75, {2 (8 t^2 γ^4 + 16 t γ^5 ϵ) CU[y, y, y, x, x] +
   <<107>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], <<1>>, 0}}
```

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -CU@x, a → -CU@a, x → -CU@y}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[SS[-T^-1 ℯ^(ℏ ϵ a) x], {a, x}],
  a → -QU@a, x → 𝕆_QU[SS[-T^-1 ℯ^(ℏ ϵ a) y], {a, y}]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}] ]
```

```
{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}
```

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}] ]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Qθ[Qθ[z]]], {z, bas}] ]
```

$$\Big\{\mathrm{QU}[y] \to \left(-1 + \frac{t\hbar}{2} - \frac{t^2\hbar^2}{8} + \frac{t^3\hbar^3}{48} - \frac{t^4\hbar^4}{384} + \frac{t^5\hbar^5}{3840}\right)\mathrm{QU}[x] + \left(-\epsilon\hbar + \frac{1}{2}t\epsilon\hbar^2 - \frac{1}{8}t^2\epsilon\hbar^3 + \frac{1}{48}t^3\epsilon\hbar^4 - \frac{1}{384}t^4\epsilon\hbar^5\right)\mathrm{QU}[a, x] + \left(-\frac{1}{2}\epsilon^2\hbar^2 + \frac{1}{4}t\epsilon^2\hbar^3 - \frac{1}{16}t^2\epsilon^2\hbar^4 + \frac{1}{96}t^3\epsilon^2\hbar^5\right)\mathrm{QU}[a, a, x] \to \mathrm{QU}[y],\ \mathrm{QU}[a] \to -\mathrm{QU}[a] \to \mathrm{QU}[a],$$

$$\mathrm{QU}[x] \to \left(-1 + \frac{t\hbar}{2} + \gamma\epsilon\hbar - \frac{t^2\hbar^2}{8} - \frac{1}{2}t\gamma\epsilon\hbar^2 - \frac{1}{2}\gamma^2\epsilon^2\hbar^2 + \frac{t^3\hbar^3}{48} + \frac{1}{8}t^2\gamma\epsilon\hbar^3 + \frac{1}{4}t\gamma^2\epsilon^2\hbar^3 - \frac{t^4\hbar^4}{384} - \frac{1}{48}t^3\gamma\epsilon\hbar^4 - \frac{1}{16}t^2\gamma^2\epsilon^2\hbar^4 + \frac{t^5\hbar^5}{3840} + \frac{1}{384}t^4\gamma\epsilon\hbar^5 + \frac{1}{96}t^3\gamma^2\epsilon^2\hbar^5\right)\mathrm{QU}[y] + \left(-\epsilon\hbar + \frac{1}{2}t\epsilon\hbar^2 + \gamma\epsilon^2\hbar^2 - \frac{1}{8}t^2\epsilon\hbar^3 - \frac{1}{2}t\gamma\epsilon^2\hbar^3 + \frac{1}{48}t^3\epsilon\hbar^4 + \frac{1}{8}t^2\gamma\epsilon^2\hbar^4 - \frac{1}{384}t^4\epsilon\hbar^5 - \frac{1}{48}t^3\gamma\epsilon^2\hbar^5\right)\mathrm{QU}[y, a] + \left(-\frac{1}{2}\epsilon^2\hbar^2 + \frac{1}{4}t\epsilon^2\hbar^3 - \frac{1}{16}t^2\epsilon^2\hbar^4 + \frac{1}{96}t^3\epsilon^2\hbar^5\right)\mathrm{QU}[y, a, a] \to \mathrm{QU}[x]\Big\}$$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

$$\mathrm{A\mathbb{D}\$f} = \frac{\gamma}{\hbar}e^{\hbar\left(\frac{t}{2}-(a+\gamma)\epsilon\right)}\left(\left(\mathrm{Cosh}\left[\hbar\left(a\epsilon + \frac{\gamma\epsilon}{2} - \frac{t}{2}\right)\right] - \mathrm{Cosh}\left[\hbar\sqrt{\left(\frac{t-\gamma\epsilon}{2}\right)^2 + \epsilon\omega}\right]\right)\Big/\left(\mathrm{Sinh}\left[\frac{\gamma\epsilon\hbar}{2}\right]\left(a^2\epsilon + a\gamma\epsilon - a t - \omega\right)\right)\right);$$

```
SS[A𝔻$f] // Simplify
```

$$\frac{1}{1440}\Big(1440 + 720\left(t - 2(a+\gamma)\epsilon\right)\hbar + 120\left(2t^2 - 7t(a+\gamma)\epsilon + \epsilon\left(7a^2\epsilon + 13a\gamma\epsilon + 6\gamma^2\epsilon + \omega\right)\right)\hbar^2 + 60\left(t^3 - 5t^2(a+\gamma)\epsilon - 2(a+\gamma)\epsilon^2\omega + t\epsilon\left(9a^2\epsilon + 17a\gamma\epsilon + 8\gamma^2\epsilon + \omega\right)\right)\hbar^3 + 2\left(6t^4 - 39t^3(a+\gamma)\epsilon - t(32a + 33\gamma)\epsilon^2\omega + 2\epsilon^2\omega^2 + t^2\epsilon\left(101a^2\epsilon + 192a\gamma\epsilon + 91\gamma^2\epsilon + 9\omega\right)\right)\hbar^4 + t\left(2t^4 - 16t^3(a+\gamma)\epsilon - t(20a + 21\gamma)\epsilon^2\omega + 2\epsilon^2\omega^2 + t^2\epsilon\left(54a^2\epsilon + 103a\gamma\epsilon + 49\gamma^2\epsilon + 4\omega\right)\right)\hbar^5\Big)$$

Scaling behaviour of A𝔻$f:

```
Simplify[A𝔻$f == ((A𝔻$f /. γ → 1) /. {ϵ → γ ϵ, a → γ^-1 a, ω → γ^-1 ω})]
```

```
True
```

```
FullSimplify[
 A𝔻$f == ((A𝔻$f /. γ → 1) /. {ℏ → γ^2 ℏ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
```

```
True
```

ADeq

```
A𝔻$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[A𝔻, QU → CU,
 {a → CU@a, x → CU@x, y → 𝕊_CU[SS[A𝔻$f], a → CU[a], ω → A𝔻$ω] ** CU@y}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[A𝔻[z1 ** z2] - A𝔻[z1] ** A𝔻[z2]]], {z1, bas}, {z2, bas}]]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

$$\mathrm{S}\mathbb{D}\$g = \sqrt{\left(\left(\mathrm{Cosh}\left[\frac{\hbar}{2}\sqrt{t^2+\gamma^2\epsilon^2+4\epsilon\varpi}\right] - \mathrm{Cosh}\left[\frac{\hbar}{2}\left(t-(2a+\gamma)\epsilon\right)\right]\right) \Big/ \left(\mathrm{Sinh}\left[\frac{\gamma\epsilon\hbar}{2}\right]\left(t(2a+\gamma) - 2a(a+\gamma)\epsilon + 2\varpi\right)\hbar / (2\gamma)\right)\right)};$$

$$\left\{\text{S}\mathbb{D}\$\text{P} = \frac{\text{Cosh}\left[\hbar\left(\frac{\epsilon-t}{2}+\epsilon\,a\right)\right]-\text{Cosh}\left[\hbar\sqrt{\frac{t^2+\epsilon^2}{4}+\epsilon\,\varpi}\right]}{\hbar\,\text{Sinh}\left[\frac{-\epsilon\,\hbar}{2}\right]\left(\varpi-\epsilon\,a^2+(t-\epsilon)\,a+t/2\right)},\right.$$

```
 Simplify[SD$P == (SD$P /. {a → -a - 1, t → -t})] // HL,
 PowerExpand@Simplify[(SD$P /. {ℏ → γ^2 ℏ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ϖ → γ^-3 ϖ}) ==
     SD$g (SD$g /. {a → -a - γ, t → -t})] // HL,
 SD$Q = Simplify[SD$P /. {a → c - 1/2}],
 Simplify[SD$Q == (SD$Q /. {c → -c, t → -t})] // HL,
 Simplify[SD$g == FullSimplify[
     √SD$Q /. c → a + 1/2 /. {ℏ → γ^2 ℏ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]] // HL
}
```

$$\left\{-\left(\left(\left(\text{Cosh}\left[\left(a\,\epsilon+\frac{1}{2}(-t+\epsilon)\right)\hbar\right]-\text{Cosh}\left[\sqrt{\frac{1}{4}(t^2+\epsilon^2)+\epsilon\,\varpi}\;\hbar\right]\right)\text{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(\frac{t}{2}+a\,(t-\epsilon)-a^2\,\epsilon+\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True},\right.$$

$$\left.-\left(\left(4\left(\text{Cosh}\left[\frac{1}{2}(t-2\,c\,\epsilon)\,\hbar\right]-\text{Cosh}\left[\frac{1}{2}\sqrt{t^2+\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\text{Csch}\left[\frac{\epsilon\,\hbar}{2}\right]\right)\Big/\left(\left(4\,c\,t+\epsilon-4\,c^2\,\epsilon+4\,\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True}\right\}$$

```
FullSimplify[SD$g /. {ℏ → γ^2 ℏ, ϵ → ϵ / γ, a → a / γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]
```

$$\sqrt{2}\sqrt{\left(\left(\gamma^2\left(-\text{Cosh}\left[\frac{1}{2}\left(t-(2\,a+\gamma^2)\,\epsilon\right)\hbar\right]+\text{Cosh}\left[\frac{1}{2}\sqrt{t^2+\gamma^4\,\epsilon^2+4\,\epsilon\,\varpi}\;\hbar\right]\right)\text{Csch}\left[\frac{1}{2}\gamma^2\,\epsilon\,\hbar\right]\right)\Big/\left(\left(t\,(2\,a+\gamma^2)-2\,a\,(a+\gamma^2)\,\epsilon+2\,\varpi\right)\hbar\right)\right)}$$

```
SS[SD$g]
```

$$1+\frac{1}{48}\left(t^2-2\,a\,t\,\epsilon-t\,\gamma\,\epsilon+2\,a^2\,\epsilon^2+2\,a\,\gamma\,\epsilon^2+2\,\epsilon\,\varpi\right)\hbar^2+\frac{1}{23\,040}\left(t^4-4\,a\,t^3\,\epsilon-2\,t^3\,\gamma\,\epsilon+16\,a^2\,t^2\,\epsilon^2+16\,a\,t^2\,\gamma\,\epsilon^2-5\,t^2\,\gamma^2\,\epsilon^2+4\,t^2\,\epsilon\,\varpi+8\,a\,t\,\epsilon^2\,\varpi+4\,t\,\gamma\,\epsilon^2\,\varpi+12\,\epsilon^2\,\varpi^2\right)\hbar^4$$

SDeq

```
SD$f = FullSimplify[e^(ℏ (t/2-ϵ a)) (SD$g /. {a → -a, t → -t})];
```

```
SS[S𝔻$f]
```

$$1+\frac{1}{2}\left(t-2\,a\,\epsilon\right)\hbar+\frac{1}{48}\left(7\,t^2-26\,a\,t\,\epsilon+t\,\gamma\,\epsilon+26\,a^2\,\epsilon^2-2\,a\,\gamma\,\epsilon^2+2\,\epsilon\,\varpi\right)\hbar^2+\frac{1}{96}\left(t-2\,a\,\epsilon\right)\left(3\,t^2-10\,a\,t\,\epsilon+t\,\gamma\,\epsilon+10\,a^2\,\epsilon^2-2\,a\,\gamma\,\epsilon^2+2\,\epsilon\,\varpi\right)\hbar^3+\frac{1}{23040}\left(121\,t^4-844\,a\,t^3\,\epsilon+62\,t^3\,\gamma\,\epsilon+2296\,a^2\,t^2\,\epsilon^2-376\,a\,t^2\,\gamma\,\epsilon^2-5\,t^2\,\gamma^2\,\epsilon^2+124\,t^2\,\epsilon\,\varpi-472\,a\,t\,\epsilon^2\,\varpi-4\,t\,\gamma\,\epsilon^2\,\varpi+12\,\epsilon^2\,\varpi^2\right)\hbar^4+\frac{1}{46080}\left(t-2\,a\,\epsilon\right)\left(33\,t^4-220\,a\,t^3\,\epsilon+22\,t^3\,\gamma\,\epsilon+584\,a^2\,t^2\,\epsilon^2-136\,a\,t^2\,\gamma\,\epsilon^2-5\,t^2\,\gamma^2\,\epsilon^2+44\,t^2\,\epsilon\,\varpi-152\,a\,t\,\epsilon^2\,\varpi-4\,t\,\gamma\,\epsilon^2\,\varpi+12\,\epsilon^2\,\varpi^2\right)\hbar^5$$

SDeq

```
S𝔻$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ CU[] / 2;
```

SDeq

```
DeclareMorphism[S𝔻, QU → CU, {a → CU@a,
  x → 𝕊CU[SS[S𝔻$f], a → CU[a], ϖ → S𝔻$ϖ] ** CU@x,
  y → 𝕊CU[SS[S𝔻$g] , a → CU[a], ϖ → S𝔻$ϖ] ** CU@y
 }]
```

Verifying the θ-symmetry:

```
Table[HL@Simplify[Cθ[S𝔻[z]] == S𝔻[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

{True, True, True}

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@Simp[S𝔻[z1 ** z2] - S𝔻[z1] ** S𝔻[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

R in QU.

Quesne's formula:

Quesne

$$\mathbb{e}_{q\_,n\_}[x\_] := \mathrm{Exp}\left[\sum_{k=1}^{n}\frac{(1-q)^k}{k\,(1-q^k)}x^k\right];\quad \mathbb{e}_{q\_}[x\_] := \mathbb{e}_{q,\$T\epsilon D}[x]$$

```
Table[Together@SeriesCoefficient[𝕖ρ,5[x], {x, 0, n}], {n, 0, 5}]
```

$$\left\{1,\,1,\,\frac{1}{1+\rho},\,\frac{1}{(1+\rho)\left(1+\rho+\rho^2\right)},\,\frac{1}{(1+\rho)^2\left(1+\rho^2\right)\left(1+\rho+\rho^2\right)},\,1\Big/\left((1+\rho)^2\left(1+\rho^2\right)\left(1+\rho+\rho^2\right)\left(1+\rho+\rho^2+\rho^3+\rho^4\right)\right)\right\}$$

```
Table[HL@FunctionExpand[QFactorial[n, ρ] SeriesCoefficient[𝕖_{ρ,5}[x], {x, 0, n}]], {n, 0, 5}]
```

{1, 1, 1, 1, 1, 1}

R

```
QU[R_{i_,j_}] := 𝕆_QU[SS[𝕖^{ℏ b₁ a₂} 𝕖_q[ℏ y₁ x₂] /. b₁ → γ^{-1} (ϵ a₁ - t_i)], {y₁, a₁} → i, {a₂, x₂} → j];
QU[R^{-1}_{i_,j_}] := S_j@QU[R_{i,j}];
```

```
QU[R_{3,4}] // Short
```

$\mathrm{QU}[\,] + \frac{\epsilon\hbar\,\mathrm{QU}[a_3, a_4]}{\gamma} + \ll 59 \gg + \mathrm{QU}[y_3, y_3, a_4, a_4, x_4, x_4]\left(\frac{\hbar^4 t_3^2}{4\gamma^2} - \frac{\epsilon\hbar^5 t_3^2}{8\gamma}\right)$

Verifying R2 (~2 secs @ $TℏD=4, $TϵD=2):

```
QU[R_{1,2} ** R^{-1}_{1,2}] // Simp // HL // Timing
```

{1.9375, QU[]}

Verifying R3 (~156 secs @ $TℏD=4, $TϵD=2):

```
{Short[lhs = QU[R_{1,2} ** R_{1,3} ** R_{2,3}]], HL@Simp[lhs - QU[R_{2,3} ** R_{1,3} ** R_{1,2}]]} // Timing
```

$\left\{2.39063, \left\{\mathrm{QU}[\,] + \frac{\epsilon\hbar\,\mathrm{QU}[a_1, a_2]}{\gamma} + \ll 347 \gg + \mathrm{QU}[y_1, a_2, x_3]\left(2\epsilon\hbar^2 + 2\epsilon\hbar^3 t_2\right) + \mathrm{QU}[y_1, x_3]\left(-\hbar^2 t_2 - \frac{1}{2}\hbar^3 t_2^2\right), 0\right\}\right\}$

𝔼

𝔼[ω, L, Q, P] means $\omega e^{L+Q} P$, where ω is a scalar, L is linear in the a's, Q is a combination of $x_i\, y_j$, and P is a perturbation polynomial. It should be interpreted via C𝕆[𝔼[...], $\{x_1, a_1, y_1\}_i$, ...] (with some default for direct interpretation), or likewise via Q𝕆[𝔼[...], $\{x_1, a_1, y_1\}_i$, ...]. In themselves, C𝕆 and Q𝕆 should have an interpretation in CU/QU by casting.